【IchigoJam】ぬるぽガッゲーム / 大石泉すきゲーム

[IchigoJam](https://ichigojam.net/) で、「ぬるぽ」と出たら素早く「ガッ」するゲームを作ってみた。
その文字のデータを差し替えて、「大石泉」が出たら素早く「すき」するゲームも作ってみた。

※IchigoJamはjig.jpの登録商標です。

# 東雲フォント (BDF) を埋め込む

今回は、Public Domain となっているビットマップフォントの「[東雲フォント](http://openlab.ring.gr.jp/efont/shinonome/)」のうち、16×16ピクセルの `shnmk16.bdf` を使用した。

このファイルでは、文字の情報が以下のような形式で記述されている。

```text:shnmk16.bdf (抜粋)
STARTCHAR 472d
ENCODING 18221
SWIDTH 960 0
DWIDTH 16 0
BBX 16 16 0 -2
BITMAP
0488
4488
2fff
1088
1088
2800
4bfe
0a22
1a22
1a22
2bfe
4a22
0a22
0a22
6bfe
1000
ENDCHAR
```

今回用いる `shnmk16.bdf` では、文字の大きさは16×16ピクセルで固定であり、`BITMAP` から `ENDCHAR` までの部分が文字の形を表しているようである。
また、`STARTCHAR` の右に書かれているのが文字を JIS で表したときの文字コードのようである。
この文字コードは、たとえば[サクラエディタ](https://sakura-editor.github.io/)で文字コードを JIS に設定し、調べたい文字の左側にキャレットを置くことで確認できる。
文字の文字コードを確認したら、その文字コードで `shnmk16.bdf` 内を検索することで、その文字のフォントが得られる。

このプログラムを実行し、`BITMAP` と `ENDCHAR` の間のデータを入力として流し込む (貼り付ける) ことで、文字の形を確認できる。
なお、描画する位置のリセットは実装していないので、画面の下まで描画したあと更に描画したい場合は、一度実行を止めて再実行することでリセットするとよい。

```text:IchigoJam BASIC
10 ' ﾌｫﾝﾄ ﾋｮｳｼﾞ
20 CLS:Y=0:X=0
30 K=INKEY()
40 IF #30<=K AND K<=#39 P=K-#30:GOTO90
50 IF #41<=K AND K<=#46 P=K-#41+10:GOTO90
60 IF #61<=K AND K<=#66 P=K-#61+10:GOTO90
70 IF K=#0A Y=Y+1:X=0
80 GOTO 30
90 FOR I=0 TO 3
100 IF P&8 DRAW X,Y
110 P=P<<1:X=X+1
120 NEXT
130 LOCATE 0,Y/2+1
140 GOTO 30
```

[OneFiveCrowd で実行する](https://onefivecrowd.mikecat.work/#text=RUN%0A&prg=CgAMJyDMq93EIMuus7zeABQADENMUzpZPTA6WD0wAAAeAApLPUlOS0VZKCkAACgAJElGICMzMDw9SyBBTkQgSzw9IzM5IFA9Sy0jMzA6R09UTzkwAAAyACZJRiAjNDE8PUsgQU5EIEs8PSM0NiBQPUstIzQxKzEwOkdPVE85MAA8ACZJRiAjNjE8PUsgQU5EIEs8PSM2NiBQPUstIzYxKzEwOkdPVE85MABGABJJRiBLPSMwQSBZPVkrMTpYPTAAUAAIR09UTyAzMAAAWgAMRk9SIEk9MCBUTyAzAGQAEElGIFAmOCBEUkFXIFgsWQAAbgAMUD1QPDwxOlg9WCsxAHgABE5FWFQAggAOTE9DQVRFIDAsWS8yKzEAjAAIR09UTyAzMA) / [IchigoJam web で実行する](https://fukuno.jig.jp/app/IchigoJam/#10%20%27%20%EF%BE%8C%EF%BD%AB%EF%BE%9D%EF%BE%84%20%EF%BE%8B%EF%BD%AE%EF%BD%B3%EF%BD%BC%EF%BE%9E%0A20%20CLS%3AY%3D0%3AX%3D0%0A30%20K%3DINKEY%28%29%0A40%20IF%20%2330%3C%3DK%20AND%20K%3C%3D%2339%20P%3DK-%2330%3AGOTO90%0A50%20IF%20%2341%3C%3DK%20AND%20K%3C%3D%2346%20P%3DK-%2341%2B10%3AGOTO90%0A60%20IF%20%2361%3C%3DK%20AND%20K%3C%3D%2366%20P%3DK-%2361%2B10%3AGOTO90%0A70%20IF%20K%3D%230A%20Y%3DY%2B1%3AX%3D0%0A80%20GOTO%2030%0A90%20FOR%20I%3D0%20TO%203%0A100%20IF%20P%268%20DRAW%20X%2CY%0A110%20P%3DP%3C%3C1%3AX%3DX%2B1%0A120%20NEXT%0A130%20LOCATE%200%2CY%2F2%2B1%0A140%20GOTO%2030%0ARUN%0A)

例として、上に掲載した文字のデータを入力すると、以下のように文字の形を確認できる。

![文字の形を確認する例](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/99656/814b6a5d-2544-bccf-ae08-4ca5c2d50530.png)

## POKE で埋め込む

使う文字のデータを、以下の変換器で IchigoJam のキャラクターパターンの形式に変換する。

<p class="codepen" data-height="600" data-default-tab="result" data-slug-hash="raBMxjm" data-pen-title="16x16 BDF converter" data-user="mike_cat" style="height: 600px; box-sizing: border-box; display: flex; align-items: center; justify-content: center; border: 2px solid; margin: 1em 0; padding: 1em;">
  <span>See the Pen <a href="https://codepen.io/mike_cat/pen/raBMxjm">
  16x16 BDF converter</a> by MikeCAT (<a href="https://codepen.io/mike_cat">@mike_cat</a>)
  on <a href="https://codepen.io">CodePen</a>.</span>
</p>
<script async src="https://cpwebassets.codepen.io/assets/embed/ei.js"></script>

<details><summary>コード</summary>

```html:HTML
<p>in</p>
<textarea id="inarea" rows="18" cols="8"></textarea>
<p>out</p>
<textarea id="outarea" rows="6" cols="50" readonly></textarea>
```

```javascript:JS
const inarea = document.getElementById("inarea");
const outarea = document.getElementById("outarea");

inarea.addEventListener("change", () => {
	const lines = inarea.value.replace(/\r\n/g, "\n").replace(/\r/g, "\n").split("\n");
	let res = "";
	for (let i = 0; i <= lines.length - 8; i += 8) {
		for (let j = 0; j < 8; j++) {
			res += parseInt(lines[i + j].substring(0, 2), 16) + ",";
		}
		for (let j = 0; j < 8; j++) {
			res += parseInt(lines[i + j].substring(2, 4), 16) + ",";
		}
	}
	outarea.value = res;
});
```

</details>

そして、それを `POKE` 命令でキャラクターパターン領域に書き込む。
(ゲームとしての実装を想定し、問題カウントと得点を想定した出力もしている)

```text:IchigoJam BASIC
10 ' ﾇﾙﾎﾟ ｶﾞｯ
20 POKE#700,0,0,16,16,19,28,16,16,0,128,128,128,240,136,132,130,41,41,70,68,74,48,0,0,2,2,2,58,68,68,58,0,0,0,16,16,17,22,24,24,0,64,64,64,240,136,132,132,41,37,66,66,77,48,0,0,2,2,2,4,4,8,48,0
30 POKE#740,0,7,0,0,1,2,7,12,0,224,64,128,0,0,224,16,16,32,0,3,4,4,3,0,8,4,4,4,136,112,192,0,0,7,0,0,0,1,3,6,0,224,32,64,128,0,240,8,8,16,0,0,0,0,1,0,4,4,4,8,16,96,128,0
40 POKE#780,0,19,16,16,16,32,35,32,6,249,25,22,16,28,240,16,32,32,32,33,42,18,17,0,16,16,16,240,28,18,224,0,0,16,16,16,16,19,32,32,22,25,25,22,30,240,16,16,32,32,32,33,42,18,17,0
50 POKE#7B8,16,16,16,240,28,18,224,0,0,1,1,1,63,1,1,1,0,2,9,5,252,8,8,8,1,2,2,4,8,16,32,0,8,8,8,8,8,16,96,0,0,0,0,0,0,0,2,17,0,0,0,0,0,0,0,16,9,8,0,0,0,0,3,0,16,16,32,32,64,128,0,0
60 VIDEO3:CLS
70 LOCATE0,0:?"11/15":LOCATE11,0:?"12345"
80 LOCATE4,4:?CHR$(#E0,#E1,#E8,#E9,#F0,#F1):LOCATE4,5:?CHR$(#E2,#E3,#EA,#EB,#F2,#F3)
90 LOCATE10,7:?CHR$(#F8,#F9,#FC,#FD):LOCATE10,8:?CHR$(#FA,#FB,#FE,#FF)
```

このプログラムは 954 バイトを消費し、1024 バイト中残りは 70 バイトである。
ゲームを実装するには、より短い表現でないといけなそうだ。

## 配列で埋め込む

`POKE` 命令は1バイトずつ書き込むため、1バイトずつ数字を書く必要があり、数値の分割や区切り文字の影響で長くなりやすい。
配列を用いることで、2バイトずつ書き込むことができ、長さを減らすことができる。
配列に書き込んだあと、`COPY` 命令を使用してキャラクターパターン領域にデータを転送する。

以下のプログラムにより、前章のキャラクターパターン形式を配列形式に変換できる。

```text:IchigoJam BASIC
10 COPY #800,#700,128
20 FOR I=0 TO 63
30 IF [I]<-9999 ?"#";HEX$([I]); ELSE ?[I];
40 IF I=31 OR I=63 ?"" ELSE ?",";
50 NEXT
```

このプログラムは、キャラクターパターンデータの前半の変換を行う。10行目の `#700` を `#780` にすることで、後半の変換を行うことができる。

変換した結果を埋め込むと、以下のコードになった。

```text:IchigoJam BASIC
10 ' ﾇﾙﾎﾟ ｶﾞｯ
20 LET[0],0,4112,7187,4112,#8000,#8080,#88F0,#8284,10537,17478,12362,0,514,14850,17476,58,0,4112,5649,6168,16384,16448,#88F0,#8484,9513,16962,12365,0,514,1026,2052,48
30 LET[32],1792,0,513,3079,-8192,#8040,0,4320,8208,768,1028,3,1032,1028,28808,192,1792,0,256,1539,-8192,16416,128,2288,4104,0,0,1,1028,2052,24592,128:COPY#700,#800,128
40 LET[0],4864,4112,8208,8227,-1786,5657,7184,4336,8224,8480,4650,17,4112,-4080,4636,224,4096,4112,4880,8224,6422,5657,-4066,4112,8224,8480,4650,17,4112,-4080,4636,224
50 LET[32],256,257,319,257,512,1289,2300,2056,513,1026,4104,32,2056,2056,4104,96,0,0,0,4354,0,0,0,4096,2057,0,0,3,4112,8224,#8040,0:COPY#780,#800,128
60 VIDEO3:CLS
70 LOCATE0,0:?"11/15":LOCATE11,0:?"12345"
80 LOCATE4,4:?CHR$(#E0,#E1,#E8,#E9,#F0,#F1):LOCATE4,5:?CHR$(#E2,#E3,#EA,#EB,#F2,#F3)
90 LOCATE10,7:?CHR$(#F8,#F9,#FC,#FD):LOCATE10,8:?CHR$(#FA,#FB,#FE,#FF)
```

このプログラムは 882 バイトを消費し、1024 バイト中残りは 142 バイトである。
`POKE` を用いた場合よりは改善したが、今回のゲームの実装には足りなかった。

## Base64の応用で埋め込む

配列を用いた埋め込みでは、2バイトを表すのに最低でも(数値1桁以上と区切り文字の)2文字を使い、(数値の桁が多く)3文字以上を使うことも多い。
[Base64](https://ja.wikipedia.org/wiki/Base64) 形式を用いれば、3バイトを4文字で表すことができるので、より効率よくデータを埋め込みやすくなる。
ただし、Base64 そのままだと文字の種類によってエンコードするデータと文字コードの関係が変化し、デコードに手間がかかってしまう。
そこで、エンコードするデータの6ビットがそのまま文字コードの下位6ビットになる文字を使うことで、簡単にデコードできるようにした。

まず、キャラクターパターンデータを設定した状態で以下のプログラムを実行し、データを取得する。

```text:IchigoJam BASIC
FORI=#700TO#7FF:?HEX$(PEEK(I),2);:NEXT:?""
```

取得したデータを CyberChef でエンコードする。

[From Hex, To Base64 - CyberChef](https://gchq.github.io/CyberChef/#recipe=From_Hex%28%27Auto%27%29To_Base64%28%27@-o0-?%27%29)

エンコードした結果を埋め込み、さらにデコーダを実装すると、以下のコードになった。

```text:IchigoJam BASIC
10 ' ﾇﾙﾎﾟ ｶﾞｯ
20 [1]="@@@PDAL\DA@@`HB@<HbD`bdiQdQJL@@@@`HBNdQDN`@@@A@PDQXXF@A@PDC0bHRDJRUBPd40@@@B@`HDA@`0@@@G@@@A@`\L@NA@`@@@8A@PH@@CA@PC@@`DA@RH\L@@@@\@@@@A@0X@8BA@`@C0B@`P@@@@@@D@A@PDBAA``@@@D1@PDB@cH@[9FQXP"
30 [2]="GO@PHB@`HRhRDP@PDAC0GAK`@@@PDA@PD2@`EadYEa;0DA@`HB@aJaHQ@A@PDO@\Dn@@@@DA@S<A@PD@@`dE?@`HB@DB@`PHDB@@B@`HB@`PX@@@@@@@@@@BDP@@@@@@@@@PBP`@@@@@@0@PDB@`PH@@@@@@":P=#700:FORI=1TO2:Q=[I]
40 C=PEEK(Q):IFC-34D=PEEK(Q+1):E=PEEK(Q+2):POKEP,C*4|D>>4&3,D<<4|E/4&15,E<<6|PEEK(Q+3)&63:P=P+3:Q=Q+4:GOTO40
50 NEXT
60 VIDEO3:CLS
70 LOCATE0,0:?"11/15":LOCATE11,0:?"12345"
80 LOCATE4,4:?CHR$(#E0,#E1,#E8,#E9,#F0,#F1):LOCATE4,5:?CHR$(#E2,#E3,#EA,#EB,#F2,#F3)
90 LOCATE10,7:?CHR$(#F8,#F9,#FC,#FD):LOCATE10,8:?CHR$(#FA,#FB,#FE,#FF)
```

このプログラムは 734 バイトを消費し、1024 バイト中残りは 290 バイトである。
この形式であれば、残りの領域で今回のゲームを実装することができた。

今回のデコーダは、簡単のため、バイト数が3の倍数 (エンコードした文字数が4の倍数) のデータにのみ対応している。
そのため、データを配列領域の最初にまで書き込むことになる。
今回のプログラムでは使わない部分なので、問題にはならない。

# ぬるぽガッゲーム

「ぬるぽ」が出たら、何かキーを押して素早く「ガッ」しよう！
でも、時々偽物が出てくるよ。
偽物のときは「ガッ」せずにスルーしよう！

```text
10 ' ﾇﾙﾎﾟ ｶﾞｯ ｹﾞｰﾑ
20 [1]="@@@PDAL\DA@@`HB@<HbD`bdiQdQJL@@@@`HBNdQDN`@@@A@PDQXXF@A@PDC0bHRDJRUBPd40@@@B@`HDA@`0@@@G@@@A@`\L@NA@`@@@8A@PH@@CA@PC@@`DA@RH\L@@@@\@@@@A@0X@8BA@`@C0B@`P@@@@@@D@A@PDBAA``@@@D1@PDB@cH@[9FQXP"
30 [2]="GO@PHB@`HRhRDP@PDAC0GAK`@@@PDA@PD2@`EadYEa;0DA@`HB@aJaHQ@A@PDO@\Dn@@@@DA@S<A@PD@@`dE?@`HB@DB@`PHDB@@B@`HB@`PX@@@@@@@@@@BDP@@@@@@@@@PBP`@@@@@@0@PDB@`PH@@@@@@":P=#700:FORI=1TO2:Q=[I]
40 C=PEEK(Q):IFC-34D=PEEK(Q+1):E=PEEK(Q+2):POKEP,C*4|D>>4&3,D<<4|E/4&15,E<<6|PEEK(Q+3)&63:P=P+3:Q=Q+4:GOTO40
50 NEXT:C=10:W=5:S=0:VIDEO3:CLS:?" 1/15          0"
60 FORQ=1TO15:LC0,0:?DEC$(Q,2):A=RND(C+W)<W:C=C-!A:W=W-A:D=RND(7)*A+A:E=D&4:F=D*2&4:G=D*4&4:WAIT60+RND(60)
70 FORI=0TO1:J=I*2:LC4,4+I:?CHR$(#E0+J+E,#E1+J+E,#E8+J+F,#E9+J+F,#F0+J+G,#F1+J+G):NEXT:CLK:CLT
80 T=TICK():IF!INKEY()IFT<120GOTO80ELSEGOTO110
90 LC10,7:?CHR$(#F8,#F9,#FC,#FD):LC10,8:?CHR$(#FA,#FB,#FE,#FF):IFAVIDEO4:S=S-600+T*5:ELSES=S+120-T
100 LC10,0:?DEC$(S,6):WAIT60+60*A
110 FORI=0TO1:LC4,4+I:?"      ":LC10,7+I:?"    ":NEXT:VIDEO3:NEXT
```

[OneFiveCrowd で実行する](https://onefivecrowd.mikecat.work/#text=RUN%0A&prg=CgAQJyDH2c7fILberyC53rDRAAAUAMJbMV09IkBAQFBEQUxcREFAQGBIQkA8SGJEYGJkaVFkUUpMQEBAQGBIQk5kUUROYEBAQEFAUERRWFhGQEFAUERDMGJIUkRKUlVCUGQ0MEBAQEJAYEhEQUBgMEBAQEdAQEBBQGBcTEBOQUBgQEBAOEFAUEhAQENBQFBDQEBgREFAUkhcTEBAQEBcQEBAQEFAMFhAOEJBQGBAQzBCQGBQQEBAQEBAREBBQFBEQkFBYGBAQEBEMUBQREJAY0hAWzlGUVhQIgAeALpbMl09IkdPQFBIQkBgSFJoUkRQQFBEQUMwR0FLYEBAQFBEQUBQRDJAYEVhZFlFYTswREFAYEhCQGFKYUhRQEFAUERPQFxEbkBAQEBEQUBTPEFAUERAQGBkRT9AYEhCQERCQGBQSERCQEBCQGBIQkBgUFhAQEBAQEBAQEBAQkRQQEBAQEBAQEBAUEJQYEBAQEBAQDBAUERCQGBQSEBAQEBAQCI6UD0jNzAwOkZPUkk9MVRPMjpRPVtJXQAAKABqQz1QRUVLKFEpOklGQy0zNEQ9UEVFSyhRKzEpOkU9UEVFSyhRKzIpOlBPS0VQLEMqNHxEPj40JjMsRDw8NHxFLzQmMTUsRTw8NnxQRUVLKFErMykmNjM6UD1QKzM6UT1RKzQ6R09UTzQwAAAyADBORVhUOkM9MTA6Vz01OlM9MDpWSURFTzM6Q0xTOj8iIDEvMTUgICAgICAgICAgMCIAPABoRk9SUT0xVE8xNTpMQzAsMDo_REVDJChRLDIpOkE9Uk5EKEMrVyk8VzpDPUMtIUE6Vz1XLUE6RD1STkQoNykqQStBOkU9RCY0OkY9RCoyJjQ6Rz1EKjQmNDpXQUlUNjArUk5EKDYwKQAARgBcRk9SST0wVE8xOko9SSoyOkxDNCw0K0k6P0NIUiQoI0UwK0orRSwjRTErSitFLCNFOCtKK0YsI0U5K0orRiwjRjArSitHLCNGMStKK0cpOk5FWFQ6Q0xLOkNMVAAAUAAsVD1USUNLKCk6SUYhSU5LRVkoKUlGVDwxMjBHT1RPODBFTFNFR09UTzExMAAAWgBgTEMxMCw3Oj9DSFIkKCNGOCwjRjksI0ZDLCNGRCk6TEMxMCw4Oj9DSFIkKCNGQSwjRkIsI0ZFLCNGRik6SUZBVklERU80OlM9Uy02MDArVCo1OkVMU0VTPVMrMTIwLVQAAGQAHkxDMTAsMDo_REVDJChTLDYpOldBSVQ2MCs2MCpBAABuAD5GT1JJPTBUTzE6TEM0LDQrSTo_IiAgICAgICI6TEMxMCw3K0k6PyIgICAgIjpORVhUOlZJREVPMzpORVhU) / [IchigoJam web で実行する](https://fukuno.jig.jp/app/IchigoJam/#10%20%27%20%EF%BE%87%EF%BE%99%EF%BE%8E%EF%BE%9F%20%EF%BD%B6%EF%BE%9E%EF%BD%AF%20%EF%BD%B9%EF%BE%9E%EF%BD%B0%EF%BE%91%0A20%20%5B1%5D%3D%22%40%40%40PDAL%5CDA%40%40%60HB%40%3CHbD%60bdiQdQJL%40%40%40%40%60HBNdQDN%60%40%40%40A%40PDQXXF%40A%40PDC0bHRDJRUBPd40%40%40%40B%40%60HDA%40%600%40%40%40G%40%40%40A%40%60%5CL%40NA%40%60%40%40%408A%40PH%40%40CA%40PC%40%40%60DA%40RH%5CL%40%40%40%40%5C%40%40%40%40A%400X%408BA%40%60%40C0B%40%60P%40%40%40%40%40%40D%40A%40PDBAA%60%60%40%40%40D1%40PDB%40cH%40%5B9FQXP%22%0A30%20%5B2%5D%3D%22GO%40PHB%40%60HRhRDP%40PDAC0GAK%60%40%40%40PDA%40PD2%40%60EadYEa%3B0DA%40%60HB%40aJaHQ%40A%40PDO%40%5CDn%40%40%40%40DA%40S%3CA%40PD%40%40%60dE%3F%40%60HB%40DB%40%60PHDB%40%40B%40%60HB%40%60PX%40%40%40%40%40%40%40%40%40%40BDP%40%40%40%40%40%40%40%40%40PBP%60%40%40%40%40%40%400%40PDB%40%60PH%40%40%40%40%40%40%22%3AP%3D%23700%3AFORI%3D1TO2%3AQ%3D%5BI%5D%0A40%20C%3DPEEK%28Q%29%3AIFC-34D%3DPEEK%28Q%2B1%29%3AE%3DPEEK%28Q%2B2%29%3APOKEP%2CC%2A4%7CD%3E%3E4%263%2CD%3C%3C4%7CE%2F4%2615%2CE%3C%3C6%7CPEEK%28Q%2B3%29%2663%3AP%3DP%2B3%3AQ%3DQ%2B4%3AGOTO40%0A50%20NEXT%3AC%3D10%3AW%3D5%3AS%3D0%3AVIDEO3%3ACLS%3A%3F%22%201%2F15%20%20%20%20%20%20%20%20%20%200%22%0A60%20FORQ%3D1TO15%3ALC0%2C0%3A%3FDEC%24%28Q%2C2%29%3AA%3DRND%28C%2BW%29%3CW%3AC%3DC-!A%3AW%3DW-A%3AD%3DRND%287%29%2AA%2BA%3AE%3DD%264%3AF%3DD%2A2%264%3AG%3DD%2A4%264%3AWAIT60%2BRND%2860%29%0A70%20FORI%3D0TO1%3AJ%3DI%2A2%3ALC4%2C4%2BI%3A%3FCHR%24%28%23E0%2BJ%2BE%2C%23E1%2BJ%2BE%2C%23E8%2BJ%2BF%2C%23E9%2BJ%2BF%2C%23F0%2BJ%2BG%2C%23F1%2BJ%2BG%29%3ANEXT%3ACLK%3ACLT%0A80%20T%3DTICK%28%29%3AIF!INKEY%28%29IFT%3C120GOTO80ELSEGOTO110%0A90%20LC10%2C7%3A%3FCHR%24%28%23F8%2C%23F9%2C%23FC%2C%23FD%29%3ALC10%2C8%3A%3FCHR%24%28%23FA%2C%23FB%2C%23FE%2C%23FF%29%3AIFAVIDEO4%3AS%3DS-600%2BT%2A5%3AELSES%3DS%2B120-T%0A100%20LC10%2C0%3A%3FDEC%24%28S%2C6%29%3AWAIT60%2B60%2AA%0A110%20FORI%3D0TO1%3ALC4%2C4%2BI%3A%3F%22%20%20%20%20%20%20%22%3ALC10%2C7%2BI%3A%3F%22%20%20%20%20%22%3ANEXT%3AVIDEO3%3ANEXT%0ARUN%0A)

![「ぬるぽ」に対して「ガッ」をした様子](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/99656/402950c0-1641-92bf-afef-1aedb48bd636.png)

* 10行目：`FILES` 対応のタイトル
* 20～30行目：フォントデータと、デコードの開始
* 40行目：フォントデータのデコードとキャラクターパターン領域への書き込み
* 50行目：出題する本物の数 `C`・偽物の数 `W`・スコア `S`・画面の初期化
* 60行目：出題処理
  * 問題カウントの表示の更新
  * 本物を出す (`A=0`) か偽物を出す (`A=1`) かの決定
  * 本物・偽物の残り数の更新
  * 偽物の場合に変える文字の決定
  * ランダムな時間待つ
* 70行目：「ぬるぽ」または偽物を表示し、時間の計測を開始する
* 80行目：キー入力およびタイムアウトの判定を行う
* 90～100行目：キーが押された時の処理
  * 90行目：「ガッ」を表示し、画面の反転と得点の減算 (偽物時) または得点の加算 (本物時) を行う
  * 100行目：得点の表示を更新し、少し待つ
* 110行目：「ぬるぽ」または偽物と「ガッ」の表示を消し、画面の反転を解除する

# 大石泉すきゲーム

「ぬるぽガッゲーム」の表示する文字だけを差し替えて、「大石泉すきゲーム」も作ってみた。
「大石泉」が表示されたら素早くキーを押して「すき」し、偽物は無視しよう。

```text
10 ' ｵｵｲｼｲｽﾞﾐ ｽｷ ｹﾞｰﾑ
20 [1]="@@@@@@A?@@B@`HB@`O>@`@DA@`HDBAA`PD@`HA@HA@L@@@@@@G<@@HBPbHRD?8B@@PDB@`PHDFA@PB@`D@`D@0@?@PDA@`HG@O<@@@@@@O0DC@0TIDPG@@PDA@PDAO0@@PDA_0HB@`P@@@C>@@@@@@\LEBQDA@\@>@`HB@`H>@@@@@<HB@<HBDB@>@`H"
30 [2]=">@`HC0@>@`PXX@G:`lRhdH2C`@@AG1HR_0@D`@C<IBS?HDPIM`@DBAA`@HRhdHbD`8B@@@@@O0@@@0Q@PG;@PDA@0@PDA@L@@@@CPDC@PDA@`@@A@PD_@@@?@@@@\HB@>D@`@@\HDA@L@0@`4C@@@@C@@@@@":P=#700:FORI=1TO2:Q=[I]
40 C=PEEK(Q):IFC-34D=PEEK(Q+1):E=PEEK(Q+2):POKEP,C*4|D>>4&3,D<<4|E/4&15,E<<6|PEEK(Q+3)&63:P=P+3:Q=Q+4:GOTO40
50 NEXT:C=10:W=5:S=0:VIDEO3:CLS:?" 1/15          0"
60 FORQ=1TO15:LC0,0:?DEC$(Q,2):A=RND(C+W)<W:C=C-!A:W=W-A:D=RND(7)*A+A:E=D&4:F=D*2&4:G=D*4&4:WAIT60+RND(60)
70 FORI=0TO1:J=I*2:LC4,4+I:?CHR$(#E0+J+E,#E1+J+E,#E8+J+F,#E9+J+F,#F0+J+G,#F1+J+G):NEXT:CLK:CLT
80 T=TICK():IF!INKEY()IFT<120GOTO80ELSEGOTO110
90 LC10,7:?CHR$(#F8,#F9,#FC,#FD):LC10,8:?CHR$(#FA,#FB,#FE,#FF):IFAVIDEO4:S=S-600+T*5:ELSES=S+120-T
100 LC10,0:?DEC$(S,6):WAIT60+60*A
110 FORI=0TO1:LC4,4+I:?"      ":LC10,7+I:?"    ":NEXT:VIDEO3:NEXT
```

[OneFiveCrowd で実行する](https://onefivecrowd.mikecat.work/#text=RUN%0A&prg=CgASJyC1tbK8sr3e0CC9tyC53rDRABQAwlsxXT0iQEBAQEBAQT9AQEJAYEhCQGBPPkBgQERBQGBIREJBQWBQREBgSEFASEFATEBAQEBAQEc8QEBIQlBiSFJEPzhCQEBQREJAYFBIREZBQFBCQGBEQGBEQDBAP0BQREFAYEhHQE88QEBAQEBATzBEQ0AwVElEUEdAQFBEQUBQREFPMEBAUERBXzBIQkBgUEBAQEM-QEBAQEBAXExFQlFEQUBcQD5AYEhCQGBIPkBAQEBAPEhCQDxIQkRCQD5AYEgiAB4AulsyXT0iPkBgSEMwQD5AYFBYWEBHOmBsUmhkSDJDYEBAQUcxSFJfMEBEYEBDPElCUz9IRFBJTWBAREJBQWBASFJoZEhiRGA4QkBAQEBATzBAQEAwUUBQRztAUERBQDBAUERBQExAQEBAQ1BEQ0BQREFAYEBAQUBQRF9AQEA_QEBAQFxIQkA-REBgQEBcSERBQExAMEBgNENAQEBAQ0BAQEBAIjpQPSM3MDA6Rk9SST0xVE8yOlE9W0ldAAAoAGpDPVBFRUsoUSk6SUZDLTM0RD1QRUVLKFErMSk6RT1QRUVLKFErMik6UE9LRVAsQyo0fEQ-PjQmMyxEPDw0fEUvNCYxNSxFPDw2fFBFRUsoUSszKSY2MzpQPVArMzpRPVErNDpHT1RPNDAAADIAME5FWFQ6Qz0xMDpXPTU6Uz0wOlZJREVPMzpDTFM6PyIgMS8xNSAgICAgICAgICAwIgA8AGhGT1JRPTFUTzE1OkxDMCwwOj9ERUMkKFEsMik6QT1STkQoQytXKTxXOkM9Qy0hQTpXPVctQTpEPVJORCg3KSpBK0E6RT1EJjQ6Rj1EKjImNDpHPUQqNCY0OldBSVQ2MCtSTkQoNjApAABGAFxGT1JJPTBUTzE6Sj1JKjI6TEM0LDQrSTo_Q0hSJCgjRTArSitFLCNFMStKK0UsI0U4K0orRiwjRTkrSitGLCNGMCtKK0csI0YxK0orRyk6TkVYVDpDTEs6Q0xUAABQACxUPVRJQ0soKTpJRiFJTktFWSgpSUZUPDEyMEdPVE84MEVMU0VHT1RPMTEwAABaAGBMQzEwLDc6P0NIUiQoI0Y4LCNGOSwjRkMsI0ZEKTpMQzEwLDg6P0NIUiQoI0ZBLCNGQiwjRkUsI0ZGKTpJRkFWSURFTzQ6Uz1TLTYwMCtUKjU6RUxTRVM9UysxMjAtVAAAZAAeTEMxMCwwOj9ERUMkKFMsNik6V0FJVDYwKzYwKkEAAG4APkZPUkk9MFRPMTpMQzQsNCtJOj8iICAgICAgIjpMQzEwLDcrSTo_IiAgICAiOk5FWFQ6VklERU8zOk5FWFQ) / [IchigoJam web で実行する](https://fukuno.jig.jp/app/IchigoJam/#10%20%27%20%EF%BD%B5%EF%BD%B5%EF%BD%B2%EF%BD%BC%EF%BD%B2%EF%BD%BD%EF%BE%9E%EF%BE%90%20%EF%BD%BD%EF%BD%B7%20%EF%BD%B9%EF%BE%9E%EF%BD%B0%EF%BE%91%0A20%20%5B1%5D%3D%22%40%40%40%40%40%40A%3F%40%40B%40%60HB%40%60O%3E%40%60%40DA%40%60HDBAA%60PD%40%60HA%40HA%40L%40%40%40%40%40%40G%3C%40%40HBPbHRD%3F8B%40%40PDB%40%60PHDFA%40PB%40%60D%40%60D%400%40%3F%40PDA%40%60HG%40O%3C%40%40%40%40%40%40O0DC%400TIDPG%40%40PDA%40PDAO0%40%40PDA_0HB%40%60P%40%40%40C%3E%40%40%40%40%40%40%5CLEBQDA%40%5C%40%3E%40%60HB%40%60H%3E%40%40%40%40%40%3CHB%40%3CHBDB%40%3E%40%60H%22%0A30%20%5B2%5D%3D%22%3E%40%60HC0%40%3E%40%60PXX%40G%3A%60lRhdH2C%60%40%40AG1HR_0%40D%60%40C%3CIBS%3FHDPIM%60%40DBAA%60%40HRhdHbD%608B%40%40%40%40%40O0%40%40%400Q%40PG%3B%40PDA%400%40PDA%40L%40%40%40%40CPDC%40PDA%40%60%40%40A%40PD_%40%40%40%3F%40%40%40%40%5CHB%40%3ED%40%60%40%40%5CHDA%40L%400%40%604C%40%40%40%40C%40%40%40%40%40%22%3AP%3D%23700%3AFORI%3D1TO2%3AQ%3D%5BI%5D%0A40%20C%3DPEEK%28Q%29%3AIFC-34D%3DPEEK%28Q%2B1%29%3AE%3DPEEK%28Q%2B2%29%3APOKEP%2CC%2A4%7CD%3E%3E4%263%2CD%3C%3C4%7CE%2F4%2615%2CE%3C%3C6%7CPEEK%28Q%2B3%29%2663%3AP%3DP%2B3%3AQ%3DQ%2B4%3AGOTO40%0A50%20NEXT%3AC%3D10%3AW%3D5%3AS%3D0%3AVIDEO3%3ACLS%3A%3F%22%201%2F15%20%20%20%20%20%20%20%20%20%200%22%0A60%20FORQ%3D1TO15%3ALC0%2C0%3A%3FDEC%24%28Q%2C2%29%3AA%3DRND%28C%2BW%29%3CW%3AC%3DC-!A%3AW%3DW-A%3AD%3DRND%287%29%2AA%2BA%3AE%3DD%264%3AF%3DD%2A2%264%3AG%3DD%2A4%264%3AWAIT60%2BRND%2860%29%0A70%20FORI%3D0TO1%3AJ%3DI%2A2%3ALC4%2C4%2BI%3A%3FCHR%24%28%23E0%2BJ%2BE%2C%23E1%2BJ%2BE%2C%23E8%2BJ%2BF%2C%23E9%2BJ%2BF%2C%23F0%2BJ%2BG%2C%23F1%2BJ%2BG%29%3ANEXT%3ACLK%3ACLT%0A80%20T%3DTICK%28%29%3AIF!INKEY%28%29IFT%3C120GOTO80ELSEGOTO110%0A90%20LC10%2C7%3A%3FCHR%24%28%23F8%2C%23F9%2C%23FC%2C%23FD%29%3ALC10%2C8%3A%3FCHR%24%28%23FA%2C%23FB%2C%23FE%2C%23FF%29%3AIFAVIDEO4%3AS%3DS-600%2BT%2A5%3AELSES%3DS%2B120-T%0A100%20LC10%2C0%3A%3FDEC%24%28S%2C6%29%3AWAIT60%2B60%2AA%0A110%20FORI%3D0TO1%3ALC4%2C4%2BI%3A%3F%22%20%20%20%20%20%20%22%3ALC10%2C7%2BI%3A%3F%22%20%20%20%20%22%3ANEXT%3AVIDEO3%3ANEXT%0ARUN%0A)

![「大石泉」に対して「すき」をした様子](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/99656/b7d6e6a3-8054-41e0-4ca3-0f1e3a55ced1.png)

# まとめ

* 東雲フォントは、JIS の文字コードと十六進数で表した文字の形の対応を記述したテキストファイルの形式で配布されている
* Base64を応用し、文字コードの下位6ビットでそのままデータの6ビットを表すエンコードを用いることで、256+2バイトのキャラクターパターンデータを効率よく (506バイトで) 埋め込み、残りの容量でゲームを実装することができた